

Heterogeneity in Dynare 7

Describing the model

Dynare Masterclass 2026

Normann Rion ¹

¹Dynare Team

July 9, 2026

Roadmap

- ❶ **Problem:** the abstract framework and Krusell–Smith.
- ❷ **Syntax:** the `.mod` file.
- ❸ **Restrictions and specifications:** parser rules.
- ❹ **Under the hood:** what the preprocessor generates.

Table of Contents

- 1 Problem
- 2 Syntax
- 3 Restrictions and specifications
- 4 Under the hood

The abstract framework: heterogeneous block

As in Bhandari et al. [2023], a continuum of agents $i \in \mathcal{I}$, discrete time.

Red: heterogeneous. Blue: aggregate.

Heterogeneous block:

$$F(\theta_{i,t}, a_{i,t-1}, x_{i,t}, \mathbb{E}_{i,t} x_{i,t+1}, Y_t) = 0.$$

where

- $x_{i,t}$: heterogeneous endogenous variables
- $a_{i,t-1} \subseteq x_{i,t-1}$: heterogeneous pre-determined variables
- $\theta_{i,t}$: heterogeneous exogenous variables
- Y_t : relevant aggregate variables
- $\dim F = \dim x_{i,t}$:

The abstract framework: aggregate block

Aggregate block $\approx$ standard RA block

Aggregate block:

$$G\left(\int \mathbf{y}_{i,t} di, A_{t-1}, X_t, \mathbb{E}_t X_{t+1}, \mathcal{E}_t\right) = 0.$$

- X_t : aggregate endogenous variables
- $A_{t-1} \subseteq X_{t-1}$: pre-determined variables.
- $\mathcal{E}_t$: aggregate i.i.d Gaussian innovations
- $\dim G = \dim X_t$

Relationship with the heterogeneous block:

- $\mathbf{y}_{i,t} \subseteq \mathbf{x}_{i,t}$: relevant aggregated heterogeneous variables
- $Y_t \subseteq [A_{t-1}, X_t, \mathbb{E}_t X_{t+1}, \mathcal{E}_t]$.
- $\theta_{i,t}$ Markovian, independent across i , independent of $\{\mathcal{E}_t\}$.

Running example: Krusell and Smith (1998)

General description

- Continuum of infinitely-lived households $i \in [0, 1]$.
- Inelastic labor supply.
- Self-insurance against idiosyncratic productivity risk, incomplete markets.
- Representative firm with Cobb–Douglas technology.
- Aggregate uncertainty: AR(1) TFP shock Z_t .
- Idiosyncratic productivity independent of Z_t (unlike Krusell and Smith [1998]).

Running example: Krusell and Smith (1998)

Household problem and FOCs

Program:

$$V(k_{i,t-1}, e_{i,t}) = \max_{c_{i,t}, k_{i,t}} \{U(c_{i,t}) + \beta \mathbb{E}_{i,t} V(k_{i,t}, e_{i,t+1})\}.$$

$$\text{s.t.} \begin{cases} U(c_{i,t}) = \frac{(c_{i,t})^{1-1/\eta}}{1-1/\eta}, k_{i,t} \geq 0 \\ e_{i,t} = \rho^e e_{i,t-1} + \epsilon_{i,t}^e \\ c_{i,t} + k_{i,t} = (1 + r_t) k_{i,t-1} + w_t \exp(e_{i,t}) \end{cases}$$

First Order Conditions $\rightarrow F$

$$k_{i,t} \geq 0 \quad \perp \quad (c_{i,t})^{-1/\eta} - \beta \mathbb{E}_{i,t} [(1 + r_{t+1}) (c_{i,t+1})^{-1/\eta}] \geq 0$$

$$c_{i,t} + k_{i,t} = (1 + r_t) k_{i,t-1} + w_t \exp(e_{i,t})$$

Running example: Krusell and Smith (1998)

Aggregate block

Aggregate optimality conditions $\rightarrow G$

$$Y_t^a = \exp(Z_t) K_{t-1}^\alpha L^{1-\alpha},$$

$$r_t = \alpha \exp(Z_t) K_{t-1}^{\alpha-1} - \delta,$$

$$w_t = (1 - \alpha) \exp(Z_t) K_{t-1}^\alpha,$$

$$K_t = \int k_{i,t} di,$$

$$\log Z_t = \rho^Z \log Z_{t-1} + (1 - \rho^Z) \log Z_{ss} + \epsilon_t^Z.$$

Mapping Krusell–Smith to the abstract framework

Heterogeneous side:

$$\begin{aligned}x_{i,t} &= [c_{i,t}, k_{i,t}]', & a_{i,t-1} &= [k_{i,t-1}]', \\ \theta_{i,t} &= [e_{i,t}]', & Y_t &= [r_t, w_t]'. \end{aligned}$$

Mapping Krusell–Smith to the abstract framework

Heterogeneous side:

$$\begin{aligned}x_{i,t} &= [c_{i,t}, k_{i,t}]', & a_{i,t-1} &= [k_{i,t-1}]', \\ \theta_{i,t} &= [e_{i,t}]', & Y_t &= [r_t, w_t]'. \end{aligned}$$

Aggregate side:

$$\begin{aligned}X_t &= [Z_t, K_t, r_t, w_t, Y_t^a]', \\ A_{t-1} &= [K_{t-1}, Z_{t-1}]', & \mathcal{E}_t &= [\epsilon_t^Z]'. \end{aligned}$$

Mapping Krusell–Smith to the abstract framework

Heterogeneous side:

$$\begin{aligned}x_{i,t} &= [c_{i,t}, k_{i,t}]', & a_{i,t-1} &= [k_{i,t-1}]', \\ \theta_{i,t} &= [e_{i,t}]', & Y_t &= [r_t, w_t]'. \end{aligned}$$

Aggregate side:

$$\begin{aligned}X_t &= [Z_t, K_t, r_t, w_t, Y_t^a]', \\ A_{t-1} &= [K_{t-1}, Z_{t-1}]', & \mathcal{E}_t &= [\epsilon_t^Z]'. \end{aligned}$$

Only $k_{i,t}$ enters G via an integral: $y_{i,t} = [k_{i,t}]'$.

Table of Contents

- 1 Problem
- 2 Syntax**
- 3 Restrictions and specifications
- 4 Under the hood

What do I declare?

Variables and shocks

- Heterogeneity dimension $\mathcal{J}$
→ `heterogeneity_dimension`.
KS: households.

What do I declare?

Variables and shocks

- Heterogeneity dimension $\mathcal{J}$
 → `heterogeneity_dimension`.
KS: households.
- Endogenous $x_{i,t}$, X_t
 → `var(heterogeneity=...), var`.
KS: $x_{i,t} = [c_{i,t}, k_{i,t}]'$, $X_t = [Z_t, K_t, r_t, w_t, Y_t^a]'$.

What do I declare?

Variables and shocks

- Heterogeneity dimension $\mathcal{J}$
 → `heterogeneity_dimension`.
KS: households.
- Endogenous $x_{i,t}$, X_t
 → `var(heterogeneity=...), var`.
KS: $x_{i,t} = [c_{i,t}, k_{i,t}]'$, $X_t = [Z_t, K_t, r_t, w_t, Y_t^a]'$.
- Individual exogenous $\theta_{i,t}$
 → `varexo(heterogeneity=...)`.
KS: $\theta_{i,t} = [e_{i,t}]'$.

What do I declare?

Variables and shocks

- Heterogeneity dimension $\mathcal{J}$
 → `heterogeneity_dimension`.
KS: `households`.
- Endogenous $x_{i,t}$, X_t
 → `var(heterogeneity=...)`, `var`.
KS: $x_{i,t} = [c_{i,t}, k_{i,t}]'$, $X_t = [Z_t, K_t, r_t, w_t, Y_t^a]'$.
- Individual exogenous $\theta_{i,t}$
 → `varexo(heterogeneity=...)`.
KS: $\theta_{i,t} = [e_{i,t}]'$.
- Aggregate innovations $\mathcal{E}_t$
 → `varexo` and the `shocks` block.
KS: $\mathcal{E}_t = [\epsilon_t^Z]'$.
- Parameters: `parameters`. No heterogeneity option yet.

Dynare .mod file

Heterogeneous variables

```
heterogeneity_dimension households;  
  
var(heterogeneity=households)  
    c (long_name = 'Consumption')  
    k (long_name = 'Assets')  
;  
  
varexo(heterogeneity=households)  
    e (long_name = 'Idiosyncratic productivity shock')  
;
```

Dynare .mod file

Aggregate variables

```
var
    Y (long_name = 'Aggregate output')
    r (long_name = 'Rate of return on capital net of depreciation')
    w (long_name = 'Wage rate')
    K (long_name = 'Aggregate capital')
    Z (long_name = 'Aggregate productivity shock')
;

varexo eps_Z (long_name = 'Aggregate productivity innovation');
```

Dynare .mod file

Parameters and aggregate innovations

parameters

```
L      (long_name = 'Labor')
alpha  (long_name = 'Share of capital in production function')
beta   (long_name = 'Subjective discount rate of households')
delta  (long_name = 'Capital depreciation rate')
eis     (long_name = 'Elasticity of intertemporal substitution')
rho_Z  (long_name = 'Aggregate TFP shock persistence')
sig_Z  (long_name = 'Aggregate TFP shock innovation std err')
Z_ss   (long_name = 'Aggregate TFP shock average value')
;
```

```
L = 1; Z_ss = 0.8816460975214567; alpha = 0.11;
beta = 0.9819527880123727; delta = 0.025; eis = 1;
```

shocks;

```
    var eps_Z; stderr 0.01;
```

```
end;
```

What do I declare?

Equations

Individual conditions:

$$F(\theta_{i,t}, a_{i,t-1}, x_{i,t}, \mathbb{E}_{i,t} x_{i,t+1}, Y_t) = 0.$$

→ `model(heterogeneity=...)` block.

What do I declare?

Equations

Individual conditions:

$$F(\theta_{i,t}, a_{i,t-1}, x_{i,t}, E_{i,t}x_{i,t+1}, Y_t) = 0.$$

→ `model(heterogeneity=...)` block.

Aggregate conditions:

$$G\left(\int y_{i,t} di, A_{t-1}, X_t, E_t X_{t+1}, \mathcal{E}_t\right) = 0.$$

→ `model` block.

What do I declare?

Equations

Individual conditions:

$$F(\theta_{i,t}, a_{i,t-1}, x_{i,t}, \mathbb{E}_{i,t} x_{i,t+1}, Y_t) = 0.$$

→ `model(heterogeneity=...)` block.

Aggregate conditions:

$$G\left(\int y_{i,t} di, A_{t-1}, X_t, \mathbb{E}_t X_{t+1}, \mathcal{E}_t\right) = 0.$$

→ `model` block.

→ `SUM(x)` $\approx \int x_{i,t} di$ operator to `sum x on the distribution of agents`. Inside aggregate equations only!

Dynare .mod file

Heterogeneous equations block

```

model(heterogeneity=households);
    [name='Euler with borrowing constraint']
    c^(-1/eis) - beta*(1+r(+1))*c(+1)^(-1/eis) = 0 [?] k >= 0;

    [name='Budget constraint']
    c + k = (1+r)*k(-1) + w*exp(e);
end;

```

Dynare .mod file

Aggregate equations block

```
model;
    [name='Production function']
    Y = exp(Z) * K(-1)^alpha * L^(1 - alpha);
    [name='Capital rental rate']
    r = alpha * exp(Z) * (K(-1) / L)^(alpha - 1) - delta;
    [name='Wage rate']
    w = (1 - alpha) * exp(Z) * (K(-1) / L)^alpha;
    [name='Capital market clearing']
    K = SUM(k);
    [name='TFP law of motion']
    log(Z) = rho_Z*log(Z(-1)) + (1-rho_Z)*log(Z_ss) + eps_Z;
end;
```


Table of Contents

- 1 Problem
- 2 Syntax
- 3 Restrictions and specifications**
- 4 Under the hood

Mixed complementarity and heterogeneous equations

Same convention as Dynare's `lmmcp` solver.

Perpendicular symbol: $\perp$ (Unicode) or `_|_` (ASCII).

Mixed complementarity and heterogeneous equations

Same convention as Dynare's `lmmcp` solver.

Perpendicular symbol: $\perp$ (Unicode) or `_|_` (ASCII).

Accepted clauses:

- $x \geq lb, x \leq ub, lb \leq x \leq ub$.
- bounds: arbitrary parameter expressions.

Mixed complementarity and heterogeneous equations

Same convention as Dynare's `lmmcp` solver.

Perpendicular symbol: $\perp$ (Unicode) or `_|_` (ASCII).

Accepted clauses:

- $x \geq lb, x \leq ub, lb \leq x \leq ub$.
- bounds: arbitrary parameter expressions.

With $RESID = LHS - RHS$ and x the bounded variable:

$$\begin{aligned} x = LB &\Rightarrow RESID(x) \geq 0, \\ LB < x < UB &\Rightarrow RESID(x) = 0, \\ x = UB &\Rightarrow RESID(x) \leq 0. \end{aligned}$$

Wrong sign?

- steady state $\rightarrow$ fails: the regime is picked from the inequality branch.
- dynamics $\rightarrow$ works: the interior branch $RESID = 0$ is sign-blind.

Implicit conditional expectations

- Standard Dynare: expectation *outside*.

$$\mathbb{E}_{i,t} F(\theta_{i,t}, a_{i,t-1}, x_{i,t}, x_{i,t+1}, \dots) = 0.$$

- Here: expectation *inside*, on heterogeneous forward-looking variables.

$$F(\theta_{i,t}, a_{i,t-1}, x_{i,t}, \mathbb{E}_{i,t} x_{i,t+1}, Y_t) = 0.$$

Implicit conditional expectations

- Standard Dynare: expectation *outside*.

$$\mathbb{E}_{i,t} F(\theta_{i,t}, a_{i,t-1}, x_{i,t}, x_{i,t+1}, \dots) = 0.$$

- Here: expectation *inside*, on heterogeneous forward-looking variables.

$$F(\theta_{i,t}, a_{i,t-1}, x_{i,t}, \mathbb{E}_{i,t} x_{i,t+1}, Y_t) = 0.$$

→ The pre-processor considers **two regimes for leads**:

- $\mathbb{E}_{i,t} x_{i,t+1}$ **stochastic**: a genuine expectation, **integrating the idiosyncratic shock kernel**. Idiosyncratic risk is priced.
- X_{t+1} **perfect foresight**: the dynamics solve for the linearized response after a one-time MIT shock. **Plain lead, no expectation operator**.
- Non-linear $\phi(x_{i,t+1})$: pre-processor introduces $z_{i,t} \equiv \phi(x_{i,t})$ and substitutes the affine lead $z_{i,t+1}$.

Non-separability of leads and lags

Heterogeneous variables: a lead $x_{i,t+1}$ and a lag $x_{i,t-1}$ entangled inside a non-separable operator are **rejected at parse time**.

- Rejected:

$$\log(x_{i,t-1} + x_{i,t+1}), \quad \exp(x_{i,t-1} \cdot x_{i,t+1}), \quad (x_{i,t-1} + x_{i,t+1})^\alpha.$$

- Admissible (separable):

$$\beta x_{i,t+1} + (1 + r_t) x_{i,t-1}, \quad x_{i,t-1} \cdot c_{i,t+1}, \quad (x_{i,t-1})^\alpha x_{i,t+1}.$$

Non-separability of leads and lags

Heterogeneous variables: a lead $x_{i,t+1}$ and a lag $x_{i,t-1}$ entangled inside a non-separable operator are **rejected at parse time**.

- Rejected:

$$\log(x_{i,t-1} + x_{i,t+1}), \quad \exp(x_{i,t-1} \cdot x_{i,t+1}), \quad (x_{i,t-1} + x_{i,t+1})^\alpha.$$

- Admissible (separable):

$$\beta x_{i,t+1} + (1 + r_t) x_{i,t-1}, \quad x_{i,t-1} \cdot c_{i,t+1}, \quad (x_{i,t-1})^\alpha x_{i,t+1}.$$

Separable: +, − always; $\times$, $\div$ when one factor carries no lead.

→ The heterogeneous lead sits under $\mathbb{E}_{i,t}$: a non-separable coupling would need a nested, second-pass expectation. Not implemented yet.

Non-separability of leads and lags

Heterogeneous variables: a **lead** $x_{i,t+1}$ and a **lag** $x_{i,t-1}$ entangled inside a non-separable operator are **rejected at parse time**.

- Rejected:

$$\log(x_{i,t-1} + x_{i,t+1}), \quad \exp(x_{i,t-1} \cdot x_{i,t+1}), \quad (x_{i,t-1} + x_{i,t+1})^\alpha.$$

- Admissible (separable):

$$\beta x_{i,t+1} + (1 + r_t) x_{i,t-1}, \quad x_{i,t-1} \cdot c_{i,t+1}, \quad (x_{i,t-1})^\alpha x_{i,t+1}.$$

Separable: +, − always; $\times$, $\div$ when one factor carries no lead.

→ The heterogeneous lead sits under $E_{i,t}$: a non-separable coupling would need a nested, second-pass expectation. Not implemented yet.

Aggregate variables escape the rule: no expectation. Non-separable couplings allowed, as standard in Dynare.

Worked example: auxiliary variable in the Euler equation

Euler written with c alone:

$$(c_{i,t})^{-1/\eta} - \beta \mathbb{E}_{i,t} \left[(1 + r_{t+1}) (c_{i,t+1})^{-1/\eta} \right] = 0 \quad \perp \quad k_{i,t} \geq 0.$$

Worked example: auxiliary variable in the Euler equation

Euler written with c alone:

$$(c_{i,t})^{-1/\eta} - \beta \mathbb{E}_{i,t} \left[(1 + r_{t+1}) (c_{i,t+1})^{-1/\eta} \right] = 0 \quad \perp \quad k_{i,t} \geq 0.$$

The heterogeneous lead $c_{i,t+1}$ enters non-linearly. The pre-processor introduces a contemporaneous auxiliary and replaces the non-linear lead by an affine one:

$$v_{i,t} \equiv (c_{i,t})^{-1/\eta}, \quad (c_{i,t})^{-1/\eta} - \beta \mathbb{E}_{i,t} \left[(1 + r_{t+1}) v_{i,t+1} \right] = 0.$$

Worked example: auxiliary variable in the Euler equation

Euler written with c alone:

$$(c_{i,t})^{-1/\eta} - \beta \mathbb{E}_{i,t} \left[(1 + r_{t+1}) (c_{i,t+1})^{-1/\eta} \right] = 0 \quad \perp \quad k_{i,t} \geq 0.$$

The heterogeneous lead $c_{i,t+1}$ enters non-linearly. The pre-processor introduces a contemporaneous auxiliary and replaces the non-linear lead by an affine one:

$$v_{i,t} \equiv (c_{i,t})^{-1/\eta}, \quad (c_{i,t})^{-1/\eta} - \beta \mathbb{E}_{i,t} \left[(1 + r_{t+1}) v_{i,t+1} \right] = 0.$$

- The lead is now affine: $(1 + r_{t+1}) v_{i,t+1}$, an aggregate factor times the heterogeneous lead $v_{i,t+1}$, separable.
- $v_{i,t}$ joins $x_{i,t}$; generated name AUX_HET_ENDO_LEAD.

Block squareness and admissible leads/lags

Squareness. Each block must be square:

$$\dim(\textcolor{red}{F}) = \dim(\textcolor{red}{x}_{i,t}), \quad \dim(\textcolor{blue}{G}) = \dim(\textcolor{blue}{X}_t).$$

Auxiliary variables count on both sides.

→ No additional equations allowed for now (e.g. Walras' law cross-checks)

Block squareness and admissible leads/lags

Squareness. Each block must be square:

$$\dim(\textcolor{red}{F}) = \dim(\textcolor{red}{x}_{i,t}), \quad \dim(\textcolor{blue}{G}) = \dim(\textcolor{blue}{X}_t).$$

Auxiliary variables count on both sides.

→ No additional equations allowed for now (e.g. Walras' law cross-checks)

Leads/lags.

- Heterogeneous endogenous: $\text{lag} \in \{-1, 0\}$, $\text{lead} \in \{0, +1\}$.
- Heterogeneous exogenous: $\text{lag} = 0$ only.
- Aggregate: standard Dynare freedom; auto auxiliary chaining for deeper lags or non-zero leads on exogenous.

Block squareness and admissible leads/lags

Squareness. Each block must be square:

$$\dim(F) = \dim(x_{i,t}), \quad \dim(G) = \dim(X_t).$$

Auxiliary variables count on both sides.

→ No additional equations allowed for now (e.g. Walras' law cross-checks)

Leads/lags.

- Heterogeneous endogenous: $\text{lag} \in \{-1, 0\}$, $\text{lead} \in \{0, +1\}$.
- Heterogeneous exogenous: $\text{lag} = 0$ only.
- Aggregate: standard Dynare freedom; auto auxiliary chaining for deeper lags or non-zero leads on exogenous.

→ Two-period-ahead heterogeneous lead? Introduce $\tilde{x}_{i,t} \equiv x_{i,t+1}$ by hand.

Specification of the exogenous shocks

Idiosyncratic shocks $\{\theta_{i,t}\}_{i,t}$ must be **independent Markov processes**:

- i.i.d. across agents i ;
- independent of $\{\mathcal{E}_t\}$.

Specification of the exogenous shocks

Idiosyncratic shocks $\{\theta_{i,t}\}_{i,t}$ must be **independent Markov processes**:

- i.i.d. across agents i ;
- independent of $\{\mathcal{E}_t\}$.

Ruled out:

- **Cross-agent correlation**
→ A shock common to all agents belongs in X_t , not $\theta_{i,t}$.
- **Idiosyncratic-aggregate correlation**

Table of Contents

- 1 Problem
- 2 Syntax
- 3 Restrictions and specifications
- 4 Under the hood**

What the preprocessor generates

Each model $M.mod$ produces a MATLAB namespace +M

What the preprocessor generates

Each model $M.mod$ produces a MATLAB namespace +M

Heterogeneous block F , prefix `dynamic_het1_` (het1: the heterogeneity dimension):

- `resid, g1, g2`: residuals, Jacobian, Hessian of F .
- `set_auxiliary_variables`: preprocessor-introduced auxiliaries.
- `complementarity_conditions`: the (LB,UB) MCP pair.

What the preprocessor generates

Each model $M.mod$ produces a MATLAB namespace +M

Heterogeneous block F , prefix `dynamic_het1_` (het1: the heterogeneity dimension):

- `resid, g1, g2`: residuals, Jacobian, Hessian of F .
- `set_auxiliary_variables`: preprocessor-introduced auxiliaries.
- `complementarity_conditions`: the (LB, UB) MCP pair.

Aggregate block G , prefix `dynamic_`.

- `resid, g1, g2`: residuals, Jacobian, Hessian of G .
→ an extra `yagg` argument feeds the cross-sectional integrals $\int y_{i,t} di$.

Auxiliary variables for complementarity

Euler equation

Euler after the v auxiliary:

$$c_{i,t}^{-1/\eta} - \beta \mathbb{E}_{i,t}[(1 + r_{t+1}) v_{i,t+1}] = 0 \quad \perp \quad k_{i,t} \geq 0.$$

Auxiliary variables for complementarity

Euler equation

Euler after the v auxiliary:

$$c_{i,t}^{-1/\eta} - \beta \mathbb{E}_{i,t}[(1 + r_{t+1}) v_{i,t+1}] = 0 \quad \perp \quad k_{i,t} \geq 0.$$

Preprocessor introduces slack $\mu_{i,t}$ (internal name MULT_L_k)

→ in `dynamic_het1_resid.m`:

$$\begin{aligned} & \left(c_{i,t}^{-1/\eta} - \beta \mathbb{E}_{i,t}[(1 + r_{t+1}) v_{i,t+1}] \right) - \mu_{i,t} = 0, \\ & k_{i,t} \cdot \mu_{i,t} = 0. \end{aligned}$$

Auxiliary variables for complementarity

Euler equation

Euler after the v auxiliary:

$$c_{i,t}^{-1/\eta} - \beta \mathbb{E}_{i,t}[(1 + r_{t+1}) v_{i,t+1}] = 0 \quad \perp \quad k_{i,t} \geq 0.$$

Preprocessor introduces slack $\mu_{i,t}$ (internal name MULT_L_k)

→ in `dynamic_het1_resid.m`:

$$\begin{aligned} (c_{i,t}^{-1/\eta} - \beta \mathbb{E}_{i,t}[(1 + r_{t+1}) v_{i,t+1}]) - \mu_{i,t} &= 0, \\ k_{i,t} \cdot \mu_{i,t} &= 0. \end{aligned}$$

→ assigned in `dynamic_het1_set_auxiliary_variables.m`:

$$\mu_{i,t} = (c_{i,t}^{-1/\eta} - \beta \mathbb{E}_{i,t}[(1 + r_{t+1}) v_{i,t+1}]) \cdot \mathbf{1}_{\{k_{i,t} \leq 0\}}.$$

Aggregated heterogeneous variables

Needed for market clearing conditions, e.g.

$$K_t = \int k_{i,t} di$$

The corresponding syntax is

$$K = \text{SUM}(k)$$

→ The pre-processor

- creates auxiliary aggregate variables (e.g. SUM_k)
- substitutes those auxiliary aggregate variables via yagg

Bibliography

Anmol Bhandari, Thomas Bourany, David Evans, and Mikhail Golosov. A perturbational approach for approximating heterogeneous agent models. Working Paper 31744, National Bureau of Economic Research, September 2023. URL <http://www.nber.org/papers/w31744>.

Per Krusell and Anthony A. Smith, Jr. Income and wealth heterogeneity in the macroeconomy. *Journal of Political Economy*, 106(5):867–896, October 1998. doi: 10.1086/250034.